VBA Code Sample #2 ( Writing an Access Contact to a Word Letter

Updated 8-9-2000

Helen Feddema (hfeddema@hvc.rr.com)

This code sample illustrates the use of OLE Automation and Word custom document properties to create a single Word letter based on a template, with data from the current address on an Access form.

Instructions

1. Copy Address 97.mdb (or Address 2000.mdb) to a location of your choice.

2. Copy DocProps.dot to \Templates\Personal Documents (create the Personal Documents subfolder if necessary).

3. Open Address 97.mdb (or Address 2000.mdb); the Contacts form will open.

4. Add a record, or edit an existing one, and click the Word Letter button.

5. A new Word letter is created, based on the DocProps template, and filled with data from the current contact on the Access form.

Code

Private Sub cmdWordLetter_Click()

'Written by Helen Feddema 4-22-98

'Last modified 7-Aug-2007

On Error GoTo ErrorHandler

   Dim appWord As Word.Application

   Dim docs As Word.Documents

   Dim strLetter As String

   Dim prps As Object

   Dim strDate As String

   Dim fso As New Scripting.FileSystemObject

   Dim fil As Scripting.File

   Dim strTemplate As String

   Dim strTemplatePath As String

   Dim strTemplateNameAndPath As String

   Dim doc As Word.Document

   Dim strTitle As String

   Dim strPrompt As String

   strDate = CStr(Date)

   'Check whether template is found in the folder

   'Get User Templates path from Word Options dialog

   '(or replace with hard-coded path for your computer)

   Set appWord = GetObject(, "Word.Application")

   strTemplatePath = appWord.Options.DefaultFilePath(wdUserTemplatesPath)

   Debug.Print "Template path: " & strTemplatePath

   strTemplatePath = strTemplatePath & "\Personal Documents\"

   strLetter = "DocProps.dot"

   strTemplateNameAndPath = strTemplatePath & strLetter

   Debug.Print "Template and path: " & strTemplateNameAndPath

On Error Resume Next

   Set fil = fso.GetFile(strTemplateNameAndPath)

   If fil Is Nothing Then

      strPrompt = "Can't find " & strLetter & " in " _

         & strTemplatePath & "; canceling"

      MsgBox strPrompt, vbCritical + vbOKOnly

      GoTo ErrorHandlerExit

   End If

On Error GoTo ErrorHandler

   Set docs = appWord.Documents

   Set doc = docs.Add(strTemplateNameAndPath)

   Set prps = doc.CustomDocumentProperties

   With prps

      .Item("TodayDate").Value = strDate

      .Item("Name").Value = Nz(Me![txtFirstName] & " " & Me![txtLastName])

      .Item("Address").Value = Nz(Me![txtAddress])

      .Item("Salutation").Value = Nz(Me![txtSalutation])

      .Item("CompanyName").Value = Nz(Me![txtCompanyName])

      .Item("City").Value = Nz(Me![txtCity])

      .Item("StateProv").Value = Nz(Me![txtStateOrProvince])

      .Item("PostalCode").Value = Nz(Me![txtPostalCode])

      .Item("JobTitle").Value = Nz(Me![txtTitle])

   End With

   With appWord

      .Visible = True

      .Activate

      .Selection.WholeStory

      .Selection.Fields.Update

      .Selection.MoveDown Unit:=wdLine, Count:=1

   End With

ErrorHandlerExit:

   Exit Sub

ErrorHandler:

   If Err = 429 Then

      'Word is not running; open Word with CreateObject

      Set appWord = CreateObject("Word.Application")

      Resume Next

   Else

      MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

      Resume ErrorHandlerExit

   End If
End Sub

Notes

· The template directory is picked up from the Word File Options dialog.

· The date is picked up from the Date() function in Access and written as text to a doc property, so it won't change every time the letter is opened.

· The Nz() function is used to prevent Nulls being passed to doc properties in case of missing address information.

· Word custom document properties are set on the Custom tab of the template's properties sheet.  The code requires the properties to be present in the Word template; if you want to adapt the code to use your own templates, either add the doc properties to your template, or copy the DocProps.dot template and modify it as needed.

Contents of Zip File

	File Name
	File Type
	Destination Folder

	Address 97.mdb
	Access 97 database
	Wherever you want

	Address 2000.mdb
	Access 2000 database
	Wherever you want

	DocProps.dot
	Word 97-2003 template
	\Templates\Personal Documents

	Wherever you want

